rewrite batch script because instead exe files I now have python.exe and bvpp.py and tg.py runners:
@echo off
echo.

if exist ..\zluda (
  set LAUNCHER=..\zluda\zluda --
  set TG_EXE=..\tg16x_z.exe
) else (
  set TG_EXE=..\tg16x.exe
)

del /Q output_tg
call time.bat START
set PROCESSOR=GPU
%LAUNCHER% %TG_EXE% input output_tg 0 ../models/TG8X_GK9_95K@7t.pth 0 100 normal jpg
echo.
if exist output_tg\0029.jpg goto :TG
set PROCESSOR=CPU
%LAUNCHER% %TG_EXE% input output_tg 0 ../models/TG8X_GK9_95K@7t.pth 7 100 normal jpg
echo.
if exist output_tg\0029.jpg goto :TG
echo TG8X failed
goto exit
:TG
echo TG8X succeeded with %PROCESSOR%
call time.bat STOP

del /Q output_bvpp
call time.bat START
%LAUNCHER% ..\bvpp8x.exe ../models/BVPP8X_st_crop333_18k.pth input output_bvpp 30 0 100 jpg
echo.
if exist output_bvpp\0029.jpg goto :BVPP
echo BVPP8X failed
goto exit
:BVPP
echo BVPP8X succeeded with GPU
call time.bat STOP

:exit
pause
----------------------
BVPP8X model usage also changed:
@echo off
chcp 65001>nul
echo.

rem BasicVSR++(bvpp8x_z.exe)を使用して8倍動画超解像を行う 
rem 8x video super-resolution using BasicVSR++(bvpp8x_z.exe).

rem OPTION------------------------------------------------------------------------------------------

rem 対応するJavPlayerのバージョン(互換性チェック用)
rem Target JavPlayer version (for compatibility check)
::JP_VER 300

rem 出力画像のズレ補正(TecoGANには0.4を指定,他のツールには通常0.0) ※1.0なら入力画像の1ピクセル分右(下)
rem Correction of output image misalignment (Normally set to 0.0. 0.4 for TecoGAN.) *With 1.0, move to the right (bottom) by one pixel of input image.
::SHIFT_X 0.0
::SHIFT_Y 0.0

rem ツールのために確保するVRAM容量(単位MB) = VRAM_USAGE + VRAM_USAGE_P*入力画像のピクセル数/10000
rem 並列実行を管理するために使用される 
rem VRAM space to be reserved for this tool (Unit:MB) = VRAM_USAGE + VRAM_USAGE_P*[Number of pixels in the input image]/10000
rem Used to manage parallel execution.
::VRAM_USAGE 500
::VRAM_USAGE_P 3200

rem ツールのために確保するメインメモリ容量(単位MB) = MEM_USAGE + MEM_USAGE_P*入力画像のピクセル数/10000
rem 並列実行を管理するために使用される 
rem MainMemory space to be reserved for this tool (Unit:MB) = MEM_USAGE + MEM_USAGE_P*[Number of pixels in the input image]/10000
rem Used to manage parallel execution.
::MEM_USAGE 600
::MEM_USAGE_P 200

rem ARGUMENTS---------------------------------------------------------------------------------------

rem 引数1(%1)は当該ツールフォルダのパス
rem argument1(%1) is the path of the relevant tool folder.

rem 引数2(%2)はワークフォルダのパス
rem argument2(%2) is the path of the wook folder.

rem 引数3(%3)は入力画像フォルダ名
rem argument3(%3) is the input folder name.

rem 引数4(%4)は出力画像フォルダ名
rem argument4(%4) is the output folder name.

rem 引数5(%5)は使用するモデルの種別(STD,AF1,etc.)
rem argument5(%5) is the type of model to use (STD,AF1,etc.)

rem 引数6(%6)は使用するGPUの番号(0<=GPUID<=3,7:CPU)
rem argument6(%6) is the ID of the GPU to use (0<=GPUID<=3,7:CPU)

rem 引数7(%7)はプロセスに割り当てられたVRAMの容量(MB)
rem argument7(%7) is VRAM capacity allocated to this process

rem 引数8(%8)はVRAMの総容量(MB)
rem argument8(%8) is total VRAM capacity

rem 引数9(%9)はポーズコマンド(エラーチェックモードの場合のみ"pause"で、通常は"")
rem argument9(%9) is the pause command. ("pause" only in error-checking mode, normally "")


rem PROCESS-----------------------------------------------------------------------------------------

:proc

cd /d %2
cd %3

rem 入力画像数をカウント 
rem Count the number of input images.
setlocal enabledelayedexpansion
set crnt=0
set step=1024
for /l %%a in (0,1,10) do (
  set /a name=crnt+step
  set name=000!name!
  set name=!name:~-4,4!
  if exist !name!.png set /a crnt=crnt+step
  set /a step=step/2
)
endlocal & set /a num_img=%crnt%+1
echo num_img=%num_img%

rem 一度に処理可能な範囲を算出 
rem Calculate the possible interval of processing at once.
set max_int=100
echo max_int=%max_int%
set /a num_div=num_img-1
set /a num_div/=max_int
set /a interval=num_img+num_div
set /a num_div+=1
set /a interval/=num_div
echo interval=%interval%

rem 使用モデルを選択 ※TG-STD→通常モデル,TG-AF*→ちらつき抑制モデル 
rem Select a model to use. *TG-STD: normal model, TG-AF*: anti-flicker model
if %5==STD (
  set bvpp_model=BVPP8X_st_3k@6f.pth
) else (
  set bvpp_model=BVPP8X_af_ft4k@6f.pth
)

rem VRAM使用割合の上限を決定 
rem Determine the upper limit of the VRAM usage percentage.
set /a vram_fraction=%7*100
set /a vram_fraction/=%8
if %vram_fraction% lss 100 (
  set /a vram_fraction+=1
)

cd /d %2

rem 超解像処理を実行 
rem Execute super-resolution processing.
:run
@echo on
"%~1\python" "%~1\bvpp.py" models\%bvpp_model% %3 %4 %interval% %6 %vram_fraction% jpg -ac
%9
@echo off

rem VRAM不足で失敗したらintervalを減らしてリトライ 
rem If it fails due to insufficient VRAM, retry with reduced interval.
set /a last_index=num_img-1
set last_index=000%last_index%
set last_index=%last_index:~-4,4%
echo last_index=%last_index%
if exist %4\%last_index%.jpg goto :exit
set /a retry+=1
if %retry% geq 55 goto :exit
set /a interval=num_img+num_div
set /a num_div+=1
set /a interval/=num_div
if %interval% leq 5 goto :exit
echo retry:%retry% (interval=%interval%)
goto :run
:exit
-------------------------
TG8X usage just remove from initial batch